[Bada Software and Hacking General] adfree:
Check this out...

I was searching again about MPRG7x30.hex via Google and in XDA...

http://www.4shared.com/file/z0PjtXvC/MPRG7x30.html?locale=de

We allready have this file... Check MD5:

http://xdaforums.com/showpost.php?p=52653884&postcount=264

But check the other files from this user...

http://www.4shared.com/file/BPGpcth4/flash.html

@echo off
setlocal ENABLEDELAYEDEXPANSION

:parse_params
IF "%1"=="" goto :done_params

IF NOT "%1"=="-s" goto :next_serial
set ___DEVICE_SERIAL=%2
shift
:next_serial

shift
goto :parse_params
:done_params

IF NOT "%___DEVICE_SERIAL%"=="" set [B]FASTBOOT[/B]_PARAMS=-s %___DEVICE_SERIAL%

@echo on

fastboot %FASTBOOT_PARAMS% flash PARTITION partition.bin flash SVN svn.bin flash DBL dbl.mbn flash OSBL osbl.mbn
fastboot %FASTBOOT_PARAMS% reboot-bootloader
@ping 127.0.0.1 -n 6 -w 1000 > nul

@REM Note: We erase APPSBL first to make sure that if flashing fails
@REM at any point, the device will reboot back into fastboot mode
fastboot %FASTBOOT_PARAMS% erase APPSBL
fastboot %FASTBOOT_PARAMS% flash FAT fat.bin flash ADSP adsp.mbn flash KERNEL boot.img flash RECOVERY_KERNEL recovery.img flash LOGO motoinitlogo.img flash CDROM cdrom.img
fastboot %FASTBOOT_PARAMS% flash SYSTEM system_part1.img flash SYSTEM system_part2.img
fastboot %FASTBOOT_PARAMS% erase MODEM_ST1 erase MODEM_ST2 erase MODEM_BKP
fastboot %FASTBOOT_PARAMS% flash MODEM_BKP NV_RBU_plus_bp_flex_modem_bkp.bin
fastboot %FASTBOOT_PARAMS% erase USERDATA
fastboot %FASTBOOT_PARAMS% erase CACHE
fastboot %FASTBOOT_PARAMS% flash APPSBL emmcboot.mbn
fastboot %FASTBOOT_PARAMS% reboot

More files possible to download... need some time...

http://www.4shared.com/file/p-0fg5ez/blank_flash.html

blank_flash.bat

@REM # (c) Copyright Motorola 2011, All rights reserved.
@REM # Motorola Confidential Proprietary
@REM # Contains confidential proprietary information of Motorola, Inc.
@REM # Reverse engineering is prohibited.
@REM # The copyright notice does not imply publication.
@REM ###################################################################################################
@REM # Revision History
@REM ###################################################################################################
@REM #
@REM # DATE AUTHOR CR NUM DESCRIPTION
@REM # ==== ====== ====== ======================================================
@REM #
@REM # 05/05/2011 fkw017 IKQCOM-992 Initial version.
@REM # 05/09/2011 fkw017 IKQCOM-1075 Support serial number & flashing bootloaders only
@REM # 06/24/2011 fkw017 IKQCOM-3026 Make blank flashing sequence more robust
@REM # 08/24/2011 far023 IKQCOM-5142 Removed qlfash.sll and devcon dependency since qflash auto detect serial port now
@REM
@REM #-------------------------------------------------------------------------------
@REM # blank_flash.bat - Flash a phone from blank state
@REM #-------------------------------------------------------------------------------

@echo off
setlocal ENABLEDELAYEDEXPANSION

@REM ------------------------------------------------------------
@REM -------------------- MAIN PROGRAM --------------------------
@REM ------------------------------------------------------------
echo Welcome to Blank Flash.
echo:

:parse_params
IF "%1"=="" goto :done_params

IF NOT "%1"=="-blonly" goto :next_bl_only
set ___BOOTLOADERS_ONLY=1
:next_bl_only

IF NOT "%1"=="-s" goto :next_serial
set ___DEVICE_SERIAL=%2
echo Using serial number %___DEVICE_SERIAL%
shift
:next_serial

shift
goto :parse_params
:done_params

IF NOT "%___DEVICE_SERIAL%"=="" set FASTBOOT_PARAMS=-s %___DEVICE_SERIAL%

call qflash -ramload MPRG7x30.hex -v

echo Sending Bootloaders...
fastboot %FASTBOOT_PARAMS% flash ALL 7x30_msimage.mbn
echo Sending Partition Table...
fastboot %FASTBOOT_PARAMS% flash PARTITION partition_mbr_only.bin
echo Rebooting...
fastboot %FASTBOOT_PARAMS% reboot-bootloader
IF "%___BOOTLOADERS_ONLY%"=="1" goto :EOF
echo Wait 2 seconds...
ping -n 3 localhost > NUL
echo Flashing all images
call flash.bat
goto :EOF

@REM ------------------------------------------------------------
@REM ------------------- FUNCTIONS BELOW ------------------------
@REM ------------------------------------------------------------

@REM ***************************************************************************
@REM ***************************************************************************
@REM
@REM CheckForRequiredFiles - Checks that the required files for flashing are present
@REM
@REM ***************************************************************************
@REM ***************************************************************************
:CheckForRequiredFiles
call qflash > NUL 2>&1
call :CheckForOneExecutable qflash.exe
call :CheckForOneExecutable fastboot.exe
call :CheckForOneFile 7x30_msimage.mbn
call :CheckForOneFile MPRG7x30.hex
call :CheckForOneFile flash.bat
goto :EOF

@REM ***************************************************************************
@REM ***************************************************************************
@REM
@REM CheckForOneFile - Checks for one file present
@REM
@REM ***************************************************************************
@REM ***************************************************************************
:CheckForOneFile
if EXIST %1 goto :EOF
set FLASHFILESNOTFOUND=1
echo ERROR: %1 not found.
goto :EOF

@REM ***************************************************************************
@REM ***************************************************************************
@REM
@REM CheckForOneExecutable - Checks to see if we can execute a file
@REM
@REM ***************************************************************************
@REM ***************************************************************************
:CheckForOneExecutable
@REM Try to execute the file to account for it being in a PATH directory.
@REM If the file doesn't exist anywhere, we'll get error 9009.
%1 > NUL 2>&1
IF NOT "%ERRORLEVEL%"=="9009" goto :EOF
set FLASHFILESNOTFOUND=1
echo ERROR: %1 not found or cannot be executed.
if NOT "%2"=="" echo One of the following files may be missing: %*
goto :EOF

Not tested nor all files downloaded... Fastboot driver etc...

Best Regards